

# RM-COBOL 85

## PrgCla.cbl

Programma "**PRGCLA.cbl**" in **RM-COBOL 85**

gestione di un generico **archivio anagrafico**

- registrazione
- variazione
- cancellazione
- ricerca
- visualizzazione
- stampa
- statistiche
- creazione iniziale

Compilatore RM-Cobol 85 versione 4.0 per Dos 2.0

**Prof. Claudio Maccherani**  
Anno scolastico 1996/1997

Esempi di VIDEATE del programma PRGCLA.CBL:

PRGCLA		Esempio RmCobol'85 - Prof. Claudio Maccherani	26/03/12
Gestione di un generico Archivio Anagrafico			
F1	-	Registrazione	F6 - Stampa
F2	-	Variazione	F7 - Statistica
F3	-	Cancellazione	F8 - ...
F4	-	Ricerca	F9 - ...
F5	-	Visualizzazione	F10 - Creaz./Azzeram.
Scegli <u>  </u> ( <b>ESC</b> fine programma )			

menu principale

PRGCLA		REGISTRAZIONE
Nominativo	Claudio	[ ESC/vuoto = fine ]
Indirizzo: Via	Cenerente	
Città	Perugia	
Provincia	PG	
Sesso	M	
Età	35_	
Peso	72_	
Altezza	17_	

registrazione

PRGCLA		RICERCA
Nominativo	Uera	
Indirizzo: Via	Senorinha de Souza	
Città	Assis	
Provincia	SP	
Sesso	F	
Età	30	
Peso	50	
Altezza	166	
[ESC fine] [PgUp primo] [PgDn succ] [Invio pref]		

ricerca

PRGCLA STAMPA

FILTRO di SELEZIONE

Ordinamento Nome/Provincia : N

Da Nome : \_\_\_\_\_

A Nome : zzzzzz\_\_\_\_\_

Provincia : \_\_\_\_\_

Sesso : \_\_\_\_\_

[C] conferma selezione

filtro di elenco/stampa

PRGCLA STATISTICA

MEDIE RISULTANTI

	Num-Tot	Età-Med	Peso-Med	Alt-Med
Maschi	3	35	74	168
Femmine	2	31	50	167
Altro	0	0	0	0
GLOBALE	5	33	64	168

[INVIO]

statistiche

PRGCLA VISUALIZZAZIONE

Nome	Via	Città	Prov	Sex	Età	Pes	Alt
Carlo	Toppo	Perugia	PG	M	33	78	170
Claudio	Cenerente	Perugia	PG	M	35	72	170
Eliane	Ranieri	Umbertide	PG	F	32	50	168
Mauro	Bruciata	Perugia	PG	M	37	73	166
Vera	Senorinha de Souza	Assis	SP	F	30	50	166

[INVIO]

elenco

Programma: PRGCLA.CBL (esempio, prof.Claudio Maccherani, Perugia, 1966)

Linguaggio: RM COBOL 85

Compilatore: versione 4.0 per DOS 2.0

```

*-----*
*   Codici tasti funzione e movimento cursore (tasto) restituiti *
*   dall'istruzione ACCEPT ... FROM ESCAPE KEY:                  *
*   01,...,20 - F1,...,F20                                       *
*   13         - Invio                                           *
*   27         - Esc                                             *
*   53         - Freccia in basso                                *
*   67         - Pagina in alto                                  *
*   68         - Pagina in basso                                 *
*   82         - Fine                                             *
*-----*
*   Colori utilizzabili nelle videate (SCREEN SECTION)          *
*   BLACK      - nero                                           *
*   BLUE       - blu                                            *
*   GREEN      - verde                                           *
*   CYAN       - celeste                                         *
*   RED        - rosso                                           *
*   MAGENTA    - viola                                           *
*   BROWN      - arancione                                       *
*   WHITE      - bianco                                           *
*-----*
*   Codici di errore di I/O (File Status) per file INDEXED      *
*   0  successo completo                                         *
*   00      - ok                                                 *
*   02      - ok, ma c'è una chiave duplicata                   *
*   04      - ok, ma lunghezza record <> lunghezza definita    *
*   05      - ok, ma ...                                         *
*   1  fine                                                     *
*   10      - READ a fine file                                    *
*   2  chiave non valida                                         *
*   21      - errore di sequenza con accesso sequenziale       *
*   22      - chiave duplicata                                    *
*   23      - record non esistente                               *
*   24      - WRITE al di fuori dei limiti massimi definiti     *
*   3  errore permanente                                         *
*   30      - errore permanente, nessuna altra informazione    *
*   35      - OPEN INPUT/I-O/EXTEND su file non esistente      *
*   37      - OPEN OUTPUT/EXTEND/I-O su file protetto in write *
*   38      - OPEN su file protetto con LOCK                   *
*   39      - errore OPEN per incompatibilità file-programma   *
*   4  errore logico                                             *
*   41      - OPEN (o DELETE FILE) per un file aperto          *
*   42      - CLOSE per un file non aperto                     *
*   43      - REWRITE dopo una READ terminata senza successo   *
*   44      - WRITE o REWRITE con record pi- corto/pi- lungo   *
*   46      - READ a fine file o dopo una START senza successo *
*   47      - READ o START su file non aperto in INPUT/I-O     *
*   48      - WRITE su file non aperto in OUTPUT/EXTEND/I-O    *
*   49      - REWRITE o DELETE su file non aperto in I-O       *
*   9  errore generale                                           *
*   93      - OPEN su un file già chiuso da un altro utente   *
*   94      - manca memoria per eseguire una OPEN              *
*   98      - difetto di struttura riscontrato nel file        *
*   99      - READ o REWRITE su file utilizzato da un altro    *
*                  utente (se non è stata definita una procedura *
*                  USE il programma aspetta che il record torni ad *
*                  essere disponibile)                          *
*-----*

```

```

*-----*
*   PRGCLA.CBL - programma di esempio - linguaggio RM COBOL 85   *
*   Prof.Claudio Maccherani - Perugia - 1996                      *
*-----*
*
*   Gestione completa di un generico archivio anagrafico
*
*   Archivio           : ANA                      File      : ARCANA.DAT
*   Organizzazione     : INDEXED                  Accesso   : DYNAMIC
*   Chiave primaria    : ANA-NOM
*   Chiave secondaria  : ANA-PRO (duplicata)
*   Record             : REC-ANA
*
*
* +-----+-----+-----+-----+
* | Campo      | Tipo | Lunghezza | Descrizione |
* +-----+-----+-----+-----+
* | ANA-NOM    | alf  | 20        | Nominativo  |
* | ANA-IND    |      |           | Indirizzo:  |
* |   ANA-VIA  | alf  | 20        | Via e numero civico |
* |   ANA-CIT  | alf  | 15        | Città       |
* |   ANA-PRO  | alf  | 2         | Provincia   |
* | ANA-SEX    | alf  | 1         | Sesso       |
* | ANA-ETA    | num  | 3         | Età         |
* | ANA-PES    | num  | 3         | Peso        |
* | ANA-ALT    | num  | 3         | Altezza     |
* +-----+-----+-----+-----+
*
*-----*

```

#### IDENTIFICATION DIVISION.

```

PROGRAM-ID.      prgcla.
AUTHOR.          Prof.Claudio Maccherani.
DATE-WRITTEN.    Anno scolastico 1995/96.
INSTALLATION.    RM-COBOL-85, Ver 4.0 for DOS 2.00+
REMARKS.         Programma esplicativo di esempio.

```

```

*-----*

```

#### ENVIRONMENT DIVISION.

##### CONFIGURATION SECTION.

```

SOURCE-COMPUTER.  rmcobol-85.
OBJECT-COMPUTER.  rmcobol-85.

```

##### INPUT-OUTPUT SECTION.

##### FILE-CONTROL.

```

*   definizione archivio anagrafico
*   SELECT ana ASSIGN TO DISK "arcana.dat"
*       ORGANIZATION      IS INDEXED
*       ACCESS MODE       IS DYNAMIC
*       RECORD KEY        IS ana-nom
*       ALTERNATE RECORD KEY IS ana-pro WITH DUPLICATES
*       FILE STATUS       IS w-fsana.
*
*   definizione file di stampa
*   SELECT sta ASSIGN TO PRINTER "PRINTER"
*       FILE STATUS       IS w-fsstta.
*
*-----*

```

\*-----

## DATA DIVISION.

### FILE SECTION.

\* definizione record archivio anagrafico

```
FD ana LABEL RECORD IS STANDARD.
01 rec-ana.
   02 ana-nom                PIC X(20) .
   02 ana-ind.
      03 ana-via             PIC X(20) .
      03 ana-cit             PIC X(15) .
      03 ana-pro             PIC X(2) .
   02 ana-sex                PIC A(1) .
   02 ana-eta                PIC 9(3) .
   02 ana-pes                PIC 9(3) .
   02 ana-alt                PIC 9(3) .
```

\* definizione record di stampa

```
FD sta LABEL RECORD IS OMITTED
   LINAGE IS 60 LINES
   LINES AT TOP      3
   LINES AT BOTTOM 3.
01 rec-sta.
   02 sta-riga          PIC X(80) .
```

### WORKING-STORAGE SECTION.

\* codici di errore file "anagrafico" e "stampa"

```
77 w-fsana                PIC X(2) .
77 w-fssta                PIC X(2) .
```

\* flag di controllo stampante accesa/spenta

```
77 w-flag                PIC A(2) .
```

\* flag di controllo record disponibile/bloccato

```
01 w-rec-lock            PIC A(2) .
88     rec-disponibile    VALUE SPACES.
88     rec-bloccato       VALUE "xx".
```

\* riga di trattini per intestazione

```
77 w-tratti              PIC X(80) VALUE
```

```
"-----"
- "-----".
```

\* carattere di tastiera acquisito come comando

```
77 w-char                PIC X(1) .
77 w-char2               PIC 9(2) .
```

\* codice del tasto funzionale digitato

```
77 w-tasto               PIC 9(2) .
```

\* scelta di menu (da 1 a 10)

```
77 w-scelta              PIC 9(2) .
```

\* descrizione opzione selezionata (per intestazione)

```
77 w-intest              PIC X(23) .
```

\* messaggio di errore e/o di avvertimento e/o di comando

```
77 w-msg                 PIC X(40) .
```

\* flag di "fine file"

```
77 w-fine                PIC A(2) .
```

\* flag di "esistenza record"

```
77 w-esiste              PIC A(2) .
```

\* modalità apertura (per riaprire il file in gestione errori)

```
77 w-open-mode           PIC A(3) .
```

\* contatori di righe e di colonne (per visualizzazione/statistica)

```
77 w-rg                  PIC 9(2) .
77 w-cl                   PIC 9(2) .
77 w-i                    PIC 9(2) .
77 w-j                    PIC 9(2) .
```

\* data di sistema

```
01 w-data.
   02 w-yy                PIC 9(2) .
```

```

02 w-mm PIC 9(2).
02 w-dd PIC 9(2).
01 w-data-vis.
02 w-dd PIC X(2).
02 FILLER PIC X(1) VALUE "/".
02 w-mm PIC X(2).
02 FILLER PIC X(1) VALUE "/".
02 w-yy PIC X(2).

* variabili per maschera selezione di ricerca
01 w-var-selezione.
02 w-nome1 PIC X(20).
02 w-nome2 PIC X(20).
02 w-provx PIC A(2).
02 w-sessox PIC A(1).
02 w-ordi PIC A(1).

* intestazione di STAMPA (con selezione)
01 w-riga-intest.
02 FILLER PIC A(23) VALUE " PRGCLA - Anagrafico - ".
02 FILLER PIC A(9) VALUE "Nomi Da: ".
02 w-r-n1 PIC X(10).
02 FILLER PIC A(4) VALUE " a: ".
02 w-r-n2 PIC X(10).
02 FILLER PIC A(10) VALUE " - Prov.: ".
02 w-r-pr PIC A(2).
02 FILLER PIC A(10) VALUE " - Sesso: ".
02 w-r-sx PIC A(1).

* riga didascalia di VISUALIZZAZIONE e/o STAMPA
01 w-riga-did.
02 w-did PIC X(80) VALUE
" Nome Via Citt...
- " Prov Sex Et... Pes Alt".

* riga anagrafica di VISUALIZZAZIONE e/o STAMPA
01 w-riga-anag.
02 FILLER PIC X(1) VALUE " ".
02 w-nome PIC X(20).
02 FILLER PIC X(1) VALUE " ".
02 w-indirizzo.
03 w-via PIC X(20).
03 FILLER PIC X(1) VALUE " ".
03 w-citta PIC X(15).
03 FILLER PIC X(1) VALUE " ".
03 w-prov PIC X(2).
03 FILLER PIC X(3) VALUE " ".
02 w-sesso PIC A(1).
02 FILLER PIC X(2) VALUE " ".
02 w-eta PIC Z(3) BLANK WHEN ZERO.
02 FILLER PIC X(1) VALUE " ".
02 w-peso PIC Z(3) BLANK WHEN ZERO.
02 FILLER PIC X(1) VALUE " ".
02 w-alt PIC Z(3) BLANK WHEN ZERO.

* matrice risultati statistici
01 w-statistica.
02 w-stat-rig OCCURS 4 TIMES.
03 w-stat OCCURS 4 TIMES PIC 9(6).

```

## SCREEN SECTION.

\* imbianca video

```
01 mask-clr.  
02 BLANK SCREEN.
```

\* menu principale

```
01 mask-menu.  
02 BLANK SCREEN  
   BACKGROUND BLUE          FOREGROUND WHITE.  
02 LINE PLUS 1 COLUMN 10 VALUE " PRGCLA " REVERSE.  
02          COLUMN 20 FOREGROUND MAGENTA  
   VALUE "Esempio RmCobol'85 - Prof.Claudio Maccherani".  
02 LINE PLUS 1 COLUMN 20 FOREGROUND CYAN  
   VALUE "Gestione di un generico Archivio Anagrafico".  
02 LINE 02          COLUMN 70 PIC X(8) FROM w-data-vis  
   FOREGROUND WHITE.  
02 LINE 04          COLUMN 01 BLANK REMAINDER  
   FOREGROUND BLACK BACKGROUND WHITE.  
02 LINE 07          COLUMN 15 VALUE " F1 - Registrazione " "  
   FOREGROUND RED.  
02 LINE PLUS 2 COLUMN 15 VALUE " F2 - Variazione " "  
02 LINE PLUS 2 COLUMN 15 VALUE " F3 - Cancellazione " "  
02 LINE PLUS 2 COLUMN 15 VALUE " F4 - Ricerca " "  
02 LINE PLUS 2 COLUMN 15 VALUE " F5 - Visualizzazione " "  
02 LINE 07          COLUMN 45 VALUE " F6 - Stampa " "  
02 LINE PLUS 2 COLUMN 45 VALUE " F7 - Statistica " "  
02 LINE PLUS 2 COLUMN 45 VALUE " F8 - ... " "  
02 LINE PLUS 2 COLUMN 45 VALUE " F9 - ... " "  
02 LINE PLUS 2 COLUMN 45 VALUE " F10 - Creaz./Azzeram. " "  
02 LINE 07          COLUMN 15 VALUE " F1 "  
   FOREGROUND BLACK BACKGROUND CYAN.  
02 LINE PLUS 2 COLUMN 15 VALUE " F2 " "  
02 LINE PLUS 2 COLUMN 15 VALUE " F3 " "  
02 LINE PLUS 2 COLUMN 15 VALUE " F4 " "  
02 LINE PLUS 2 COLUMN 15 VALUE " F5 " "  
02 LINE 07          COLUMN 45 VALUE " F6 " "  
02 LINE PLUS 2 COLUMN 45 VALUE " F7 " "  
02 LINE PLUS 2 COLUMN 45 VALUE " F8 " "  
02 LINE PLUS 2 COLUMN 45 VALUE " F9 " "  
02 LINE PLUS 2 COLUMN 45 VALUE " F10" "  
02 LINE PLUS 3 COLUMN 25 FOREGROUND BLACK BACKGROUND WHITE  
   VALUE " Scegli ( ESC fine programma )".  
02          COLUMN 38 VALUE "ESC"  
   FOREGROUND BLACK BACKGROUND CYAN.  
02          COLUMN 33 PIC 9(2) TO w-char2 BELL  
   FOREGROUND WHITE BACKGROUND BLUE.  
02 LINE 24          COLUMN 01 BLANK REMAINDER  
   FOREGROUND WHITE BACKGROUND BLUE.
```

\* intestazione comune alle varie funzioni

```
01 mask-intest.  
02 BLANK SCREEN          FOREGROUND WHITE BACKGROUND BLUE.  
02 LINE 01 COLUMN 01 BLANK LINE          BACKGROUND WHITE.  
02 LINE 02 COLUMN 01 BLANK LINE.  
02 LINE 03 COLUMN 01 BLANK LINE.  
02 LINE 24 COLUMN 01 BLANK REMAINDER.  
02 LINE 02 COLUMN 10 VALUE " PRGCLA "  
   FOREGROUND WHITE BACKGROUND BLUE.  
02 LINE 02 COLUMN 20 PIC X(23) FROM w-intest  
   FOREGROUND RED BACKGROUND CYAN.  
02 LINE 04 COLUMN 01 BLANK LINE  
   FOREGROUND WHITE BACKGROUND BLUE.
```

\* maschera di display messaggio e di acquisizione comando

```
01 mask-msg-1.  
02 LINE 24 COLUMN 10 PIC X(40) FROM w-msg
```



```

                                FOREGROUND BLUE BACKGROUND WHITE.
02 LINE 24 COLUMN 60 PIC X(1)  TO w-char AUTO.

01 mask-msg-2.
02 LINE 24 COLUMN 10 VALUE " " BLANK REMAINDER
                                FOREGROUND BLUE BACKGROUND WHITE.
02 LINE 23 COLUMN 01 VALUE " "
                                FOREGROUND WHITE BACKGROUND BLUE.

*   maschera anagrafica (didascalie) e richiesta nominativo
01 mask-ana-nome.
02 LINE 06          COLUMN 10 VALUE "Nominativo"          BLANK LINE.
02 LINE PLUS 2 COLUMN 10 VALUE "Indirizzo: Via" BLANK LINE.
02 LINE PLUS 1 COLUMN 10 VALUE "          Citt..." BLANK LINE.
02 LINE PLUS 1 COLUMN 10 VALUE "          Provincia" BLANK LINE.
02 LINE PLUS 2 COLUMN 10 VALUE "Sesso" BLANK LINE.
02 LINE PLUS 2 COLUMN 10 VALUE "Et..." BLANK LINE.
02 LINE PLUS 2 COLUMN 10 VALUE "Peso" BLANK LINE.
02 LINE PLUS 2 COLUMN 10 VALUE "Altezza" BLANK LINE.
02 LINE 06          COLUMN 30 PIC X(20) USING ana-nom.

*   maschera anagrafica registraz.e/o variaz. e/o display (input/output)
01 mask-ana-campi.
02 LINE 06          COLUMN 30 PIC X(20) FROM ana-nom HIGHLIGHT.
02 LINE PLUS 2 COLUMN 30 PIC X(20) USING ana-via HIGHLIGHT.
02 LINE PLUS 1 COLUMN 30 PIC X(15) USING ana-cit HIGHLIGHT.
02 LINE PLUS 1 COLUMN 30 PIC X(2)  USING ana-pro HIGHLIGHT.
02 LINE PLUS 2 COLUMN 30 PIC A(1)  USING ana-sex HIGHLIGHT.
02 LINE PLUS 2 COLUMN 30 PIC 9(3)  USING ana-eta HIGHLIGHT.
02 LINE PLUS 2 COLUMN 30 PIC 9(3)  USING ana-pes HIGHLIGHT.
02 LINE PLUS 2 COLUMN 30 PIC 9(3)  USING ana-alt HIGHLIGHT.

*   maschera di selezione per visualizzazione/stampa
01 mask-selezione.
02 LINE 06 COLUMN 20 VALUE " FILTRO di SELEZIONE "
                                FOREGROUND BLACK BACKGROUND BROWN.
02 LINE 08 COLUMN 20 VALUE "Ordinamento Nome/Provincia :"
                                FOREGROUND WHITE BACKGROUND BLUE.
02 LINE PLUS 2 COLUMN 20 VALUE "Da Nome      :".
02 LINE PLUS 2 COLUMN 20 VALUE "A Nome      :".
02 LINE PLUS 2 COLUMN 20 VALUE "Provincia   :".
02 LINE PLUS 2 COLUMN 20 VALUE "Sesso       :".
02 LINE 08          COLUMN 50 PIC A(1)  USING w-ordi  HIGHLIGHT.
02 LINE PLUS 2 COLUMN 35 PIC X(20) USING w-nome1  HIGHLIGHT.
02 LINE PLUS 2 COLUMN 35 PIC X(20) USING w-nome2  HIGHLIGHT.
02 LINE PLUS 2 COLUMN 35 PIC X(2)  USING w-provx  HIGHLIGHT.
02 LINE PLUS 2 COLUMN 35 PIC A(1)  USING w-sessox  HIGHLIGHT.

*   maschera didascalie per visualizzazione
01 mask-visualizzazione.
02 LINE 04 COLUMN 01 BLANK REMAINDER.
02 LINE 04 COLUMN 01 PIC X(80) FROM w-riga-did.
02 LINE 05 COLUMN 01 PIC X(80) FROM w-tratti.

*   maschera di visualizzazione risultati statistica
01 mask-statistica.
02 LINE 04 COLUMN 01 BLANK REMAINDER.
02 LINE 06 COLUMN 20 VALUE " MEDIE RISULTANTI "
                                FOREGROUND BLACK BACKGROUND BROWN.
02 LINE 08 COLUMN 20 FOREGROUND WHITE BACKGROUND BLUE
                                VALUE "Num-Tot Et...-Med Peso-Med Alt-Med".
02 LINE 10 COLUMN 10 VALUE "Maschi".
02 LINE 12 COLUMN 10 VALUE "Femmine".
02 LINE 14 COLUMN 10 VALUE "Altro".
02 LINE 16 COLUMN 10 VALUE "GLOBALE".

```

\*-----

\*-----

## PROCEDURE DIVISION.

\* gestione errori

## DECLARATIVES.

```
err-ana SECTION. USE AFTER ERROR PROCEDURE ON ana.
* gestione errori i/o file "ana"
err-ana-x.
    IF w-fsana = "35" THEN
*       se il file non esiste (dopo OPEN INPUT/I-O) viene creato
        DISPLAY "File ANAGRAFICO NON esistente"
            LINE 20 POSITION 10
        OPEN OUTPUT ana
        CLOSE ana
        DISPLAY "File ANAGRAFICO CREATO - [INVIO]"
            LINE 21 POSITION 10
        ACCEPT w-char LINE 21 POSITION 45
        DISPLAY " " LINE 20 POSITION 1 ERASE EOL
        DISPLAY " " LINE 21 POSITION 1 ERASE EOL
        IF w-open-mode = "INP" THEN
            OPEN INPUT ana
        END-IF
        IF w-open-mode = "I-O" THEN
            OPEN I-O ana
        END-IF
    ELSE
        IF w-fsana = "99" THEN
*       tentativo di leggere (in I-O) un record per varia. e/o canc.
*       un record bloccato (lock) da un altro utente (in I-O)
            MOVE "xx" TO w-rec-lock
        ELSE
*       tutti gli altri errori vengono segnalati ed il programma termina
            DISPLAY "Errore " w-fsana " su file ANAGRAFICO "
            STOP RUN
        END-IF
    END-IF.
ex-err-ana-x. EXIT.

err-sta SECTION. USE AFTER ERROR PROCEDURE ON sta.
* gestione errori stampante
err-sta-x.
    IF w-fssta = "30" THEN
        DISPLAY
            "Stampante NON collegata o spenta. R-riprova A-annulla"
            LINE 20 POSITION 1
        MOVE " " TO w-char
        PERFORM UNTIL w-char = "R" OR w-char = "r" OR
            w-char = "A" OR w-char = "a"
            ACCEPT w-char LINE 20 POSITION 65
        END-PERFORM
        DISPLAY " " LINE 20 POSITION 1 ERASE EOL
        IF w-char = "R" OR w-char = "a" THEN
            OPEN OUTPUT sta
            MOVE "ok" TO w-flag
        ELSE
            MOVE "no" TO w-flag
        END-IF
    ELSE
        DISPLAY "Errore " w-fssta " su file di STAMPA "
        STOP RUN
    END-IF.
ex-err-sta-x. EXIT.

END DECLARATIVES.
```

**inizio-programma SECTION.**

\* menu principale -----

**inizio.**

```
*    acquisizione data di sistema
ACCEPT w-data FROM DATE
MOVE CORR w-data TO w-data-vis
*    ciclo principale di menu
PERFORM UNTIL w-tasto = 27
    DISPLAY mask-menu
    ACCEPT  mask-menu
    ACCEPT w-tasto FROM ESCAPE KEY
    IF w-char2 NOT < 1 AND w-char2 NOT > 10 THEN
        MOVE w-char2 TO w-tasto
    END-IF
    MOVE w-tasto TO w-scelta
    EVALUATE w-scelta
        WHEN 1 THRU 3  PERFORM reg-var-can
        WHEN 4          PERFORM ricerca
        WHEN 5 THRU 6  PERFORM sel-vis-sta
        WHEN 7          PERFORM statistica
        WHEN 10        PERFORM crea
        WHEN 27
            DISPLAY "Programma Terminato"
            LINE 20 POSITION 30
    END-EVALUATE
END-PERFORM.
STOP RUN.
```

**ex-inizio.**

\* F1 / F2 / F3 : reg.e/o var.e/o can. record anagrafico -----

**reg-var-can.**

```
MOVE SPACES TO rec-ana
IF w-scelta = 1 THEN
    MOVE " REGISTRAZIONE " TO w-intest
ELSE
    IF w-scelta = 2 THEN
        MOVE " VARIAZIONE " TO w-intest
    ELSE
        MOVE " CANCELLAZIONE " TO w-intest
    END-IF
END-IF
DISPLAY mask-intest
MOVE "I-O" TO w-open-mode
OPEN I-O ana
MOVE SPACES TO rec-ana
PERFORM UNTIL w-tasto = 27
    DISPLAY mask-ana-campi
    DISPLAY mask-ana-nome
    DISPLAY "[ ESC/vuoto = fine ]" LINE 6 POSITION 60 LOW
    ACCEPT  mask-ana-nome
    ACCEPT w-tasto FROM ESCAPE KEY
    IF ana-nom = SPACES THEN
        MOVE 27 TO w-tasto
    END-IF
    IF w-tasto NOT = 27 THEN
        PERFORM leggi-record
        IF w-esiste = "si" AND w-scelta = 1 THEN
            DISPLAY mask-ana-campi
            MOVE "Record Esistente" TO w-msg
            PERFORM comando
        ELSE
            IF w-esiste = "no" AND
                ( w-scelta = 2 OR w-scelta = 3 ) THEN
                MOVE "Record NON Esistente" TO w-msg
                PERFORM comando
            ELSE

```

```

        IF w-scelta = 1 THEN
            PERFORM registra
        ELSE
            IF w-scelta = 2 THEN
                PERFORM varia
            ELSE
                IF w-scelta = 3 THEN
                    PERFORM cancella
                END-IF
            END-IF
        END-IF
    END-IF
END-IF
MOVE SPACES TO rec-ana
END-PERFORM.
CLOSE ana
MOVE 0 TO w-tasto.
ex-reg-var-can. EXIT.

* F1 : registrazione record anagrafico -----
registra.
    DISPLAY mask-ana-campi
    ACCEPT mask-ana-campi
    ACCEPT w-tasto FROM ESCAPE KEY
    MOVE "[R] registra" TO w-msg
    PERFORM comando
    IF w-char = "R" OR w-char = "r" THEN
        WRITE rec-ana
            INVALID KEY DISPLAY "Record esistente" BEEP
                                LINE 22 POSITION 45
                                ACCEPT w-char LINE 22 POSITION 70
        END-WRITE
    END-IF.
ex-registra. EXIT.

* F2 : variazione record anagrafico -----
varia.
    DISPLAY mask-ana-campi
    ACCEPT mask-ana-campi
    ACCEPT w-tasto FROM ESCAPE KEY
    MOVE "[R] registra" TO w-msg
    PERFORM comando
    IF w-char = "R" OR w-char = "r" THEN
        REWRITE rec-ana
            INVALID KEY DISPLAY "Record NON esistente" BEEP
                                LINE 22 POSITION 45
                                ACCEPT w-char LINE 22 POSITION 70
        END-REWRITE
    END-IF.
ex-varia. EXIT.

* F3 : cancellazione record anagrafico -----
cancella.
    DISPLAY mask-ana-campi
    MOVE "[C] cancella" TO w-msg
    PERFORM comando
    IF w-char = "C" OR w-char = "c" THEN
        DELETE ana
            INVALID KEY DISPLAY "Errore Cancellazione" BEEP
                                LINE 22 POSITION 45
                                ACCEPT w-char LINE 22 POSITION 70
        END-DELETE
    END-IF.
ex-cancella. EXIT.

```

```

* F4 : ricerca di un record anagrafico -----
ricerca.
    MOVE SPACES      TO rec-ana
    MOVE " RICERCA " TO w-intest
    DISPLAY mask-intest
    MOVE "INP" TO w-open-mode
    OPEN INPUT ana
    MOVE "no" TO w-esiste
    PERFORM UNTIL w-tasto = 27
        DISPLAY
            "[ESC fine]  [PgUp primo]  [PgDn succ]  [Invio pref]"
            LINE 22 POSITION 10
        DISPLAY mask-ana-nome
        DISPLAY mask-ana-campi
        ACCEPT  mask-ana-nome
        ACCEPT  w-tasto FROM ESCAPE KEY
        IF w-tasto = 13 AND ana-nom NOT = SPACES THEN
            PERFORM ric-prefisso
        ELSE
            IF w-tasto = 67 THEN
                PERFORM ric-primo
            ELSE
                IF w-tasto = 68 THEN
                    PERFORM ric-successivo
                END-IF
            END-IF
        END-IF
        IF w-esiste = "si" THEN
            DISPLAY mask-ana-campi
        END-IF
    END-PERFORM.
    CLOSE ana
    MOVE 0 TO w-tasto.
ex-ricerca. EXIT.

* lettura dato il prefisso -----
ric-prefisso.
    START ana KEY IS NOT < ana-nom
        INVALID KEY MOVE "no" TO w-esiste
        NOT INVALID KEY MOVE "si" TO w-esiste
    END-START
    IF w-esiste = "si" THEN
        READ ana NEXT RECORD
    END-IF.
ex-ric-prefisso. EXIT.

* lettura del primo -----
ric-primo.
    MOVE SPACES TO ana-nom
    START ana KEY IS NOT < ana-nom
        INVALID KEY MOVE "no" TO w-esiste
        NOT INVALID KEY MOVE "si" TO w-esiste
    END-START
    IF w-esiste = "si" THEN
        READ ana NEXT RECORD
    END-IF.
ex-ric-primo. EXIT.

* lettura del successivo -----
ric-successivo.
    IF w-esiste = "si" THEN
        READ ana NEXT RECORD
        AT END MOVE "no" TO w-esiste
        NOT AT END MOVE "si" TO w-esiste
    END-READ
    END-IF.
ex-ric-successivo. EXIT.

```

```

* F5 / F6 : visualizzazione e/o stampa -----
sel-vis-sta.
    MOVE "ok" TO w-flag
    MOVE "INP" TO w-open-mode
    OPEN INPUT ana
    IF w-scelta = 6 THEN
        OPEN OUTPUT sta
    END-IF
    IF w-scelta = 5 THEN
        MOVE " VISUALIZZAZIONE " TO w-intest
    ELSE
        IF w-scelta = 6 THEN
            MOVE " STAMPA " TO w-intest
        END-IF
    END-IF
    IF w-flag = "ok" THEN
*       w-flag viene messo a "no" dalla gestione errori (DECLARATIVES)
*       se la stampante è spenta
        MOVE SPACES TO w-var-selezione
        MOVE "zzzzz" TO w-nome2
        MOVE "N" TO w-ordi
        DISPLAY mask-intest
        DISPLAY mask-selezione
        ACCEPT mask-selezione
        ACCEPT w-tasto FROM ESCAPE KEY
        IF w-tasto NOT = 27
            MOVE "[C] conferma selezione" TO w-msg
            PERFORM comando
            IF w-char = "C" OR w-char = "c"
                MOVE 6 TO w-rg
                IF w-scelta = 5 THEN
                    PERFORM vis-intest
                ELSE
                    IF w-scelta = 6 THEN
                        PERFORM sta-intest
                    END-IF
                END-IF
            END-IF
            MOVE w-nome1 TO ana-nom
            MOVE w-provx TO ana-pro
            IF w-ordi = "N" THEN
*               segue la chiave primaria "ana-nom"
                START ana KEY IS NOT < ana-nom
                    INVALID KEY MOVE "si" TO w-fine
                    NOT INVALID KEY MOVE "no" TO w-fine
                END-START
            ELSE
*               segue la chiave secondaria "provincia"
                START ana KEY IS NOT < ana-pro
                    INVALID KEY MOVE "si" TO w-fine
                    NOT INVALID KEY MOVE "no" TO w-fine
                END-START
            END-IF
            IF w-fine = "no" THEN
                READ ana NEXT RECORD
*               controlla se terminare per supero dei limiti
                IF ( w-ordi = "N" AND ana-nom > w-nome2 ) OR
                    ( w-ordi = "P" AND w-provx NOT = SPACES
                        AND ana-pro > w-provx )
                    MOVE "si" TO w-fine
                END-IF
            END-IF
*               ciclo lettura records anagrafici per vis./sta.
            PERFORM ciclo-vis-sta UNTIL w-fine = "si"
            IF w-scelta = 5 THEN
                MOVE "[INVIO]" TO w-msg
                PERFORM comando
            ELSE

```

```

                IF w-scelta = 6 THEN
                    PERFORM sta-salto-pagina
                END-IF
            END-IF
        END-IF
    END-IF
END-IF
CLOSE ana.
IF w-scelta = 6 AND w-flag = "ok" THEN
    CLOSE sta
END-IF
MOVE 0 TO w-tasto.
ex-sel-vis-sta. EXIT.

* ciclo lettura records per visualizzazione e/o stampa -----
ciclo-vis-sta.
*   controllo filtro di selezione
    IF ( w-sessox = SPACE OR w-sessox = ana-sex ) AND
        ( w-provx = SPACES OR w-provx = ana-pro ) AND
        ( w-nome1 NOT > ana-nom AND ana-nom NOT > w-nome2 ) THEN
        MOVE ana-nom TO w-nome
        MOVE ana-via TO w-via
        MOVE ana-cit TO w-citta
        MOVE ana-pro TO w-prov
        MOVE ana-sex TO w-sesso
        MOVE ana-eta TO w-eta
        MOVE ana-pes TO w-peso
        MOVE ana-alt TO w-alt
        IF w-scelta = 5 THEN
            DISPLAY w-riga-anag LINE w-rg POSITION 1
            IF w-rg = 20 THEN
                MOVE "[INVIO]" TO w-msg
                PERFORM comando
                MOVE 6 TO w-rg
                PERFORM vis-intest
            ELSE
                ADD 1 TO w-rg
            END-IF
        ELSE
            WRITE rec-sta FROM w-riga-anag
        END-IF
    END-IF
    READ ana NEXT RECORD
        AT END MOVE "si" TO w-fine
        NOT AT END MOVE "no" TO w-fine
    END-READ.
*   controlla se terminare per supero dei limiti
    IF ( w-ordi = "N" AND ana-nom > w-nome2 ) OR
        ( w-ordi = "P" AND w-provx NOT = SPACES
          AND ana-pro > w-provx )
        MOVE "si" TO w-fine
    END-IF.
ex-ciclo-vis-sta. EXIT.

* intestazione per visualizzazione -----
vis-intest.
    DISPLAY mask-visualizzazione.
ex-vis-intest. EXIT.

* intestazione per stampa -----
sta-intest.
    MOVE w-nome1 TO w-r-n1
    MOVE w-nome2 TO w-r-n2
    MOVE w-provx TO w-r-pr
    MOVE w-sessox TO w-r-sx
    WRITE rec-sta FROM w-tratti AFTER ADVANCING 1 LINES
    WRITE rec-sta FROM w-riga-intest

```

```

WRITE rec-sta FROM w-tratti
WRITE rec-sta FROM w-riga-did AFTER ADVANCING 2 LINES
WRITE rec-sta FROM w-tratti.
ex-sta-intest. EXIT.

* salto pagina per stampa -----
sta-salto-pagina.
  MOVE SPACES TO rec-sta
  WRITE rec-sta BEFORE ADVANCING PAGE.
ex-sta-salto-pagina. EXIT.

* F7 : statistica -----
*
*      numero      et...      peso      altezza
*      W-STAT      1          2          3          4
*
*      +-----+-----+-----+-----+
*      maschi  1  |         |         |         |
*      +-----+-----+-----+-----+
*      femmine 2  |         |         |         |
*      +-----+-----+-----+-----+
*      altro   3  |         |         |         |
*      +-----+-----+-----+-----+
*      totale  4  |         |         |         |
*      +-----+-----+-----+-----+
*
*      statistica.
  MOVE "INP" TO w-open-mode
  OPEN INPUT ana
  MOVE " STATISTICA " TO w-intest
  DISPLAY mask-intest
  MOVE "[C] conferma selezione" TO w-msg
  PERFORM comando
  IF w-char = "C" OR w-char = "c"
    DISPLAY "Attendere - Elaborazione in corso ... "
      LINE 10 POSITION 10 HIGH BLINK
    MOVE ZEROES TO w-statistica
    MOVE SPACES TO ana-nom
    START ana KEY IS NOT < ana-nom
      INVALID KEY MOVE "si" TO w-fine
      NOT INVALID KEY MOVE "no" TO w-fine
    END-START
    IF w-fine = "no" THEN
      READ ana NEXT RECORD
    END-IF
  *
  *      ciclo lettura records anagrafici per statistica
  PERFORM UNTIL w-fine = "si"
    IF ana-sex = "M" OR ana-sex = "m" THEN
      MOVE 1 TO w-i
    ELSE
      IF ana-sex = "F" OR ana-sex = "f" THEN
        MOVE 2 TO w-i
      ELSE
        MOVE 3 TO w-i
      END-IF
    END-IF
    ADD 1          TO w-stat(w-i , 1)
    ADD ana-eta TO w-stat(w-i , 2)
    ADD ana-pes TO w-stat(w-i , 3)
    ADD ana-alt TO w-stat(w-i , 4)
    READ ana NEXT RECORD
    AT END MOVE "si" TO w-fine
  END-READ
  END-PERFORM
  *
  *      totalizzazioni sull'ultima riga della matrice
  PERFORM VARYING w-i FROM 1 BY 1 UNTIL w-i > 3
    AFTER w-j FROM 1 BY 1 UNTIL w-j > 4
      ADD w-stat(w-i , w-j) TO w-stat(4 , w-j)
  END-PERFORM
  *
  *      calcolo delle medie

```



```

        PERFORM VARYING w-i FROM 1 BY 1 UNTIL w-i > 4
        IF w-stat(w-i , 1) > 0
            COMPUTE w-stat(w-i , 2) =
                w-stat(w-i , 2) / w-stat(w-i , 1)
            COMPUTE w-stat(w-i , 3) =
                w-stat(w-i , 3) / w-stat(w-i , 1)
            COMPUTE w-stat(w-i , 4) =
                w-stat(w-i , 4) / w-stat(w-i , 1)
        END-IF
    END-PERFORM
*   visualizzazione risultati statistica
    DISPLAY mask-statistica
    PERFORM VARYING w-i FROM 1 BY 1 UNTIL w-i > 4
        AFTER w-j FROM 1 BY 1 UNTIL w-j > 4
            COMPUTE w-rg = 8 + ( w-i * 2 )
            COMPUTE w-cl = 15 + ( w-j * 9 )
            DISPLAY w-stat(w-i , w-j) HIGH CONVERT
                LINE w-rg POSITION w-cl
    END-PERFORM
END-IF
MOVE "[INVIO]" TO w-msg
PERFORM comando
CLOSE ana.
MOVE 0 TO w-tasto.
ex-statistica. EXIT.

* F10 : creazione / azzeramento archivio anagrafico -----
crea.
    MOVE " CREAZIONE/AZZERAMENTO " TO w-intest
    DISPLAY mask-intest
    DISPLAY "ATTENZIONE. L'archivio, se esiste, sar... AZZERATO"
        LINE 5 POSITION 10
    MOVE "[C] conferma creazione" TO w-msg
    PERFORM comando
    IF w-char = "C" OR w-char = "c" THEN
        MOVE "OUT" TO w-open-mode
        OPEN OUTPUT ana
        CLOSE ana
        MOVE "ARCHIVIO CREATO" TO w-msg
        PERFORM comando
    END-IF.
    MOVE 0 TO w-tasto.
ex-crea. EXIT.

* lettura record anagrafico -----
* NB: non c'e'bisogno di "MOVE xxx TO ana-nom" perché, il campo "ana-nom"
*     usato nelle maschere fa riferimento al campo "ana-nom" del record
*     per cui acquisendo la "mask-ana-nome" la chiave "ana-nom" risulta
*     inizializzata
*     si fa il ciclo per controllare il LOCK del record da parte di altri
*     utenti - se il record è bloccato "w-flag" vale "xx" (DECLARATIVES)
leggi-record.
    MOVE SPACES TO w-rec-lock
    PERFORM WITH TEST AFTER UNTIL rec-disponibile
        READ ana KEY IS ana-nom
            INVALID KEY MOVE "no" TO w-esiste
            NOT INVALID KEY MOVE "si" TO w-esiste
    END-READ
    IF rec-bloccato THEN
*       il record è attualmente bloccato da un altro utente
        MOVE SPACES TO w-fsana
        MOVE "Record Bloccato - [R] ritenta [A] annulla"
            TO w-msg
        PERFORM comando
        IF w-char = "A" OR w-char = "a" THEN
            STOP RUN
        END-IF
    
```

```
        END-IF
    END-PERFORM.
    ex-leggi-record. EXIT.

* display messaggio ed acquisizione comando -----
comando.
    DISPLAY mask-msg-1.
    ACCEPT  mask-msg-1.
    DISPLAY mask-msg-2.
    MOVE SPACES TO w-msg.
ex-comando. EXIT.
```